Chapitre 1
GEOMETRIE AFFINE ALGEBRIQUE

Nous introduisons dans ce chapitre les notions de base de géométrie affine et euclidienne et leur adaptation au calcul formel :

- *alignement, coordonnées et coordonnées barycentriques,*
- *parallélisme, intersection de droites,*
- *produit scalaire, applications affines, isométries et problèmes de distances.*

Les applications développées en dimension 2 et 3 couvrent un très large spectre de la géométrie enseignée au lycée et dans le premier cycle supérieur.

I. ESPACES AFFINES ET EUCLIDIENS FORMELS

Il existe plusieurs façons de considérer un espace affine en calcul formel, en particulier sous Mathematica suivant la représentation d'un point de cet espace :

(1) Une représentation formelle pure, qui ne tient pas compte de la dimension finie ou infinie de l'espace affine.
(2) La représentation avec les coordonnées barycentriques, la notion de barycentre constitue un pilier de la géométrie affine.
(3) La représentation d'un point dans un espace de dimension finie, muni d'un repère affine par la liste de ses coordonnées : il s'agit du cadre de la géométrie analytique classique. Cette manière de procéder nécessite réflexion pour le choix des repères.
(4) La représentation par un nombre complexe dans un plan affine euclidien.

---

1. Espaces affines formels

Soit $\mathcal{E}$ un espace affine associé à un espace vectoriel E, de dimension finie ou infinie.

1.1 Définition des objets élémentaires en géométrie formelle

Nous faisons abstraction de la dimension, de la notion de repère.
• Le vecteur $\overrightarrow{AB}$ est représenté formellement par l'objet $B - A$, de type `Symbol`, associé à la relation de Chasles $\overrightarrow{AB} = \overrightarrow{OB} - \overrightarrow{OA}$.
• Le milieu P du segment $[AB]$ est défini géométriquement par l'égalité vectorielle $\overrightarrow{OP} = \frac{1}{2}\left(\overrightarrow{OA} + \overrightarrow{OB}\right)$. Formellement, nous lui associons l'objet $\frac{1}{2}(A + B)$, de type `Symbol`.
• Le centre de gravité G du triangle ABC est caractérisé géométriquement par la relation $\overrightarrow{OG} = \frac{1}{3}\left(\overrightarrow{OA} + \overrightarrow{OB} + \overrightarrow{OC}\right)$, donc formellement par $G = \frac{1}{3}(A + B + C)$.
• Les identificateurs C, D, I, $T\ldots$ sont réservés, la première lettre de l'identificateur de chaque point sera systématiquement une minuscule, sage précaution sous Mathematica.

```
Clear[vect, milieu, centreGravite]
vect[a_, b_] := b - a
milieu[a_, b_] := (a + b) / 2
centreGravite[a_, b_, c_] := (a + b + c) / 3
```

Ces fonctions sont facultatives, on peut remplacer *milieu(a,b)* par $\frac{a+b}{2}$ sans perte de lisibilité.

Un quadrilatère $A\,B\,C\,D$ est un parallélogramme si et seulement si $\overrightarrow{A\,B} = \overrightarrow{D\,C}$.
Un quadrilatère formel (a, b, c, d) est donc un parallélogramme si et seulement si
$b - a = c - d$ ou encore $a - b + c - d = 0$.

```
Clear[parallelogrammeQ]
parallelogrammeQ[a_, b_, c_, d_] := a - b + c - d == 0
```

1.2 Colinéarité et alignement en géométrie formelle

■ Fonctions-tests

Ecrivons deux fonctions pour tester :
- la colinéarité de deux vecteurs
- l'alignement de trois points.

Pour déterminer si deux vecteurs $\vec{u}$ et $\vec{v}$ sont colinéaires, testons le caractère numérique de la variable symbolique $\frac{u}{v}$ en supposant $v \neq 0$.

```
Clear[colineairesQ]
colineairesQ[u_, v_] := NumericQ[Simplify[u / v]]
```

Il est très important de noter que dans cette représentation les points sont des objets symboliques et non de type `List`. Les deux vecteurs $\vec{u}(-2, -4, -6)$, $\vec{v}(1, 2, 3)$ sont géométriquement colinéaires, pourtant ce n'est pas ce qu'affiche la fonction *colineairesQ*.

```
colineairesQ[{-2, -4, -6}, {1, 2, 3}]

False
```

L'explication tient au format de la liste $\frac{u}{v} = \{-2, -2, -2\}$, non numérique dans cet exemple. Si de plus, l'une des composantes du vecteur $\vec{v}$ était nulle, le quotient des listes u et v serait non défini et Mathematica renverrait un avertissement.
Limiter l'application de la fonction aux objets de type `Symbol` apporterait une mauvaise réponse, le milieu d'un segment étant de type `Times`.

Les points A, B, C sont alignés si et seulement si les vecteurs $\overrightarrow{A\,B}$ et $\overrightarrow{A\,C}$ sont colinéaires.

```
Clear[alignesQ]
alignesQ[a_, b_, c_] := colineairesQ[vect[a, b], vect[a, c]]
```

■ Applications

Dans le cadre de la géométrie formelle "pure" :

- les figures ne jouent pas un rôle important
- de nombreuses affectations sont évitées.

Considérons un triangle ABC *de centre de gravité* G *et les trois médianes* $[AM]$, $[BN]$, $[CP]$, *les points* M, N, P *étant les milieux respectifs des côtés* $[BC]$, $[AC]$, $[AB]$.
Vérifier les propriétés :

- $\overrightarrow{PN} = \frac{1}{2}\overrightarrow{BC}$.
- $\overrightarrow{AM} + \overrightarrow{BN} + \overrightarrow{CP} = \vec{0}$.
- *L'alignement des triplets de points* (A, G, M), (B, G, N), (C, G, P).
- $\overrightarrow{GP} = \frac{1}{3}\overrightarrow{CP}$, $\overrightarrow{GN} = \frac{1}{3}\overrightarrow{BN}$, $\overrightarrow{GM} = \frac{1}{3}\overrightarrow{AM}$.
- *Les triangles* ABC *et* MNP *ont le même centre de gravité.*




```
Simplify[vect[milieu[a, b], milieu[a, c]] == 1/2 vect[b, c]]
```

```
True
```

```
Simplify[vect[a, m] + vect[b, n] + vect[c, p],
  {m == milieu[b, c], n == milieu[a, c], p == milieu[a, b]}]
```

```
0
```

```
g := centreGravite[a, b, c]
{alignesQ[a, g, milieu[b, c]],
 alignesQ[b, g, milieu[c, a]], alignesQ[c, g, milieu[a, b]]}
```

```
{True, True, True}
```

```
Simplify[(milieu[b, c] - g) / (milieu[b, c] - a)]
```

$\frac{1}{3}$

Nous venons de prouver que les médianes d'un triangle se coupent en un point situé au tiers de chaque médiane par rapport aux côtés.
Vérifions que les triangles $A\,B\,C$ et $M\,N\,P$ ont le même centre de gravité :

```
centreGravite[a, b, c] == Simplify[
  centreGravite[milieu[b, c], milieu[c, a], milieu[a, b]]]
```

```
True
```

A, B, C sont trois points d'un espace affine. Les points D, E, F, G sont définis par :

$\overrightarrow{B\,D} = \frac{1}{3}\,\overrightarrow{B\,A}$ $\qquad \overrightarrow{C\,E} = \frac{1}{4}\,\overrightarrow{C\,B}$ $\qquad \overrightarrow{C\,F} = \frac{3}{4}\,\overrightarrow{C\,B}$ $\qquad \overrightarrow{A\,G} = \frac{8}{9}\,\overrightarrow{A\,E}$

• *Démontrer que les droites* $(A\,E)$ *et* $(D\,F)$ *sont parallèles.*
• *Vérifier que les points C, D, G sont alignés.*

Etudions la colinéarité des vecteurs $\overrightarrow{A\,E}$ et $\overrightarrow{D\,F}$ en évaluant sous conditions l'expression $\frac{e-a}{f-d}$.

$$\text{Simplify}\left[\frac{e-a}{f-d},\ \left\{d-b==\frac{a-b}{3},\ e-c==\frac{b-c}{4},\ f-c==\frac{3\,(b-c)}{4}\right\}\right]$$

```
3
```

Donc $\overrightarrow{A\,E} = 3\,\overrightarrow{D\,F}$, les droites $(A\,E)$ et $(D\,F)$ sont parallèles.

Etudions l'alignement des points C, D, G :

```
Clear[g]
```

$$\text{Simplify}\left[\text{alignesQ}[c,\ d,\ g],\ \left\{d-b==\frac{a-b}{3},\ e-c==\frac{b-c}{4},\ f-c==\frac{3\,(b-c)}{4},\ g-a==\frac{8\,(e-a)}{9}\right\}\right]$$

```
False
```

La réponse de Mathematica est fausse, il réalise une simplification partielle, en conséquence, la relation d'alignement est non satisfaite. Le calcul formel tend des pièges!

Il faut procéder différemment, en réalisant les affectations concernant les points D, E, F, G. Ce que l'on perd en élégance se convertit en efficacité.

```
Clear[a, b, c]
d := b + (a - b)/3; e := c + (b - c)/4;
f := c + 3 (b - c)/4; g := a + 8 (e - a)/9;
{Simplify[alignesQ[c, d, g]], Simplify[(g - c) / (d - c)]}
```

$$\left\{\text{True}, \frac{1}{3}\right\}$$

Il en résulte que $\overrightarrow{CG} = \frac{1}{3}\,\overrightarrow{CD}$, les trois points G, C, D sont alignés.

Démontrons que dans un quadrilatère
• *les quatre droites joignant un sommet au centre de gravité du triangle formé des trois autres sommets,*
• *les deux droites joignant les milieux des côtés opposés,*
• *la droite joignant les milieux des diagonales*
sont concourantes.



Vérifions que toutes les droites passent par le centre de gravité G du quadrilatère.

```
Clear[a, b, c, d, g]
g := (a + b + c + d) / 4
{alignesQ[milieu[a, b], milieu[c, d], g],
 alignesQ[a, centreGravite[b, c, d], g],
 alignesQ[milieu[a, c], milieu[b, d], g]}
```

```
{True, True, True}
```

Considérons un parallélépipède $A\,B\,C\,D\,E\,F\,G\,H$.
• Démontrer que les plans $(B\,D\,E)$ et $(C\,F\,H)$ sont parallèles.
• Soient I et J les centres de gravité respectifs des triangles $B\,D\,E$ et $C\,F\,H$.
Démontrer que les points A, I, J, G sont alignés et que $\overrightarrow{A\,I} = \overrightarrow{I\,J} = \overrightarrow{J\,G}$.



Définissons D tel que $\overrightarrow{A\,B} = \overrightarrow{D\,C}$ ou encore $d = a - b + c$. Le quadrilatère $E\,F\,G\,H$ se déduit de $A\,B\,C\,D$ dans la translation de vecteur $\vec{u}$, ce que l'on écrit : $(e, f, g, h) = u + (a, b, c, d)$.
Testons la nature des quadrilatères $E\,B\,C\,H$ et $E\,D\,C\,F$:

```
Clear[a, b, c, d, e, f, g, h, u]
d := a - b + c; {e, f, g, h} = u + {a, b, c, d};
Map[Apply[parallelogrammeQ, #] &, {{e, b, c, h}, {e, d, c, f}}]
```

```
{True, True}
```

Les quadrilatères $E\,B\,C\,H$ et $E\,D\,C\,F$ sont deux parallèlogrammes, $\overrightarrow{E\,B} = \overrightarrow{H\,C}$ et $\overrightarrow{E\,D} = \overrightarrow{F\,C}$.
Les droites $(E\,B)$ et $(H\,C)$ sont parallèles, les droites $(E\,D)$ et $(F\,C)$ sont parallèles.
Le plan $(E\,B\,D)$ contient donc un couple de droites sécantes parallèles à un couple de droites sécantes du plan $(C\,F\,H)$. Les deux plans $(B\,D\,E)$ et $(C\,F\,H)$ sont parallèles.

Vérifions l'alignement des points A, I, G, puis celui de A, J, G :

```
i := centreGravite[b, d, e]
j := centreGravite[c, f, h]
{alignesQ[a, i, g], alignesQ[a, j, g]}
```

```
{True, True}
```

Comparons les vecteurs $\overrightarrow{A\,I}$, $\overrightarrow{I\,J}$, $\overrightarrow{J\,G}$, ce qui redémontre l'alignement de A, I, J, G.

```
Simplify[a - i == i - j == j - g]
```

```
True
```

1.3 Propriétés équivalentes en géométrie formelle

Ecrivons une importante fonction pour tester l'équivalence de deux propriétés :

```
Clear[equivalenceQ]
equivalenceQ[p_, q_] := And[Simplify[p, q], Simplify[q, p]]
```

Démontrons que le quadrilatère $A\,B\,C\,D$ est un parallélogramme si et seulement si les diagonales $[A\,C]$ et $[B\,D]$ ont le même milieu :

```
equivalenceQ[parallelogrammeQ[a, b, c, d],
 milieu[a, c] == milieu[b, d]]
```

```
True
```

$A\,B\,C$ et $A_1\,B_1\,C_1$ sont deux triangles de centres de gravité respectifs G et G_1.
Démontrer que $G = G_1$ si et seulement si il existe un point D tel que (D, B, A_1, C) et (D, B_1, A, C_1) soient deux parallélogrammes.